
Zusammenfassung 

Mit der Graphics Template Language und passenden Optionen lassen sich mühelos ani-
mierte Graphiken erstellen, mit denen z.B. ein zeitlicher Verlauf dargestellt werden kann. 
An Hand eines einfachen Beispiels, in dem die Anzahl unerwünschter Ereignisse über die 
Wochen einer klinischen Studie, getrennt nach Behandlung und Altersklasse, angezeigt 
werden, wird dies präsentiert. Es wird zunächst mit PROC TEMPLATE ein Prototyp für 
den zu erstellenden Plot jeder Studienwoche definiert. Hierbei können beliebige GTL-
Befehle verwendet werden. Die Animation findet durch Aneinanderreihung der einzelnen 
Graphiken jeder einzelnen Woche in einer gif-Datei statt, wobei die Woche als BY-
Variable in PROC SGRENDER auftaucht. Die notwendigen Optionen und ODS-Befehle 
zur Erstellung einer solchen animierten Graphik werden angegeben. 

 
Schlüsselwörter: GTL, Grafik, Animation 

1 Motivation 

Seit in SAS Graphiken auch mit der Graphics Template Language (GTL) erstellt wer-
den können, lassen sich eine Vielzahl verschiedener Graphiken schnell programmieren. 
Aber auch die Möglichkeiten, spezielle Wünsche an die Gestaltung der Graphiken zu 
stellen, sind erheblich vielfältiger geworden. Neben statischen Graphiken kann es (z.B. 
bei der Darstellung von Zeitverläufen o.ä.) auch sinnvoll sein, mehrere Bilder animiert 
in einer Graphik zu präsentieren. Das Erstellen animierter Graphiken ist durch Verwen-
dung entsprechender GTL-Befehle und ODS-Optionen problemlos möglich. 

2 GTL – Graphics Template Language 

Die Möglichkeiten zur Erstellung von Graphiken sind seit der Einführung von GTL er-
heblich erweitert worden. Sowohl die schnelle Erstellung von Graphiken als auch die 
anspruchsvolle Gestaltung von Graphiken sind möglich. Grundsätzlich muss zunächst 
in einem PROC TEMPLATE Schritt die allgemeine Struktur und Zusammensetzung der 
gewünschten Graphik definiert werden. Hier wird das Aussehen der Graphik, also die 
Anordnung verschiedener Elemente (Plots, Bars, Achsen, etc.) vorgegeben. Mit PROC 
SGRENDER kann dann eine Graphik unter Verwendung eines vorher definierten 
Templates und passenden Daten erstellt werden. 
Die Definition eines Layouts hat (in einem einfachen Beispiel) folgendes Aussehen: 

   PROC TEMPLATE; 
     DEFINE STATGRAPH mygraphs.scatter; 
       LAYOUT OVERLAY; 
         SCATTERPLOT X=height Y=weight; 
       ENDLAYOUT; 
     END; 
   RUN; 
 
Es wird ein Scatterplot mit der x-Variablen HEIGHT und der y-Variablen WEIGHT 
definiert. 
Vor der Graphikerstellung mit PROC SGRENDER können verschiedene globale Ein-
stellungen für die folgenden Graphiken mit ODS-Befehlen festgelegt werden: 
 
   ODS LISTING GPATH=“c:\documents\graphics“; 
   ODS GRAPHICS / IMAGEFMT=png; 
 
   PROC SGRENDER DATA=sashelp.class TEMPLATE=mygraphs.sctter; 
   RUN; 
 
   ODS GRAPHICS / RESET=all; 
 
Nach Vorgabe des Pfades, in dem die zu erstellenden Graphiken gespeichert werden 
sollen, und dem Angabe des gewünschten Ausgabeformats, wird das vordefinierte Lay-
out aufgerufen und auf die Daten aus SASHELP.CLASS angewendet. 
Das allgemeine Prinzip bleibt bei der Graphikerstellung mittels GTL immer gleich, die 
zu benutzenden Befehle und Optionen sind jedoch sehr vielfältig und erlauben nahezu 
das Erstellen verschiedenster Graphiktypen. Genauere Details können in der SAS-Hilfe 
gefunden werden. 

3 Animierte Graphiken 

3.1 Die Syntax 

Mit animierten Graphiken können mehrere Bilder hintereinander in einer Graphik ange-
zeigt werden. Dies kann zum Beispiel bei Zeitverläufen helfen, diese besser zu veran-
schaulichen. Bei Verwendung von GTL-Befehlen werden animierte Graphiken mit eini-
gen speziellen Optionen und dem ODS PRINTER Befehl erzeugt. Die allgemeine Syn-
tax hat folgendes Aussehen: 
 
  OPTIONS PRINTERPATH=gif ANIMATION=start ANIMDURATION=0.5  
          ANIMLOOP=yes NOANIMOVERLAY; 
  ODS PRINTER file="c:\documents\graphics\animation.gif"; 
 
  PROC TEMPLATE; 
    DEFINE STATGRAPH <template-name>; 
      BEGINGRAPH / <options>; 

        <GTL statements>; 
      ENDGRAPH; 
    END; 
  RUN; 
 
  PROC SGRENDER DATA=<data-set-name> TEMPLATE=<template-name>; 
  RUN; 
 
  OPTIONS PRINTERPATH=gif ANIMATION=stop; 
  ODS PRINTER close; 

 
Es wird eine aus mehreren einzelnen Graphiken bestehende gif-Datei erstellt, bei deren 
Aufruf diese einzelnen Graphiken hintereinander dargestellt werden. ANIMATI-
ON=start beginnt mit der Erstellung der Animation. Mit der Option ANIMDURATION 
wird festgelegt, wie lange jede einzelne Graphik angezeigt werden soll. ANIMLOOP 
legt fest, ob die Animation ständig wiederholt wird oder eben nicht. Die Option 
NOANIMOVERLAY bestimmt, dass die einzelnen Graphiken innerhalb der Animation 
sequentiell hintereinander angezeigt werden. Nach Erstellung und Darstellung der ani-
mierten Graphiken muss mit ANIMATION=stop die Animation beendet werden. Um 
für eine Animation mehrere Graphiken zu erzeugen, kann in PROC SGRENDER ein 
BY-Befehl verwendet werden oder es können auch mehrere PROC SGRENDER Aufru-
fe erstellt werden. 

3.2 Beispieldaten 

Um an einem Beispiel die Erstellung einer animierten Graphik zu erläutern, verwenden 
wir einen einfachen klinischen Beispieldatensatz, der für zwei unterschiedliche Behand-
lungen (Verum, Placebo) die Anzahl an Unerwünschten Ereignissen (Adverse Events – 
AEs), getrennt nach 8 Altersklassen, über die Wochen 0 bis 26 enthält. Der Verlauf der 
Anzahl an AEs, unterschieden nach Altersklassen und Behandlungen, soll über die Zeit 
animiert dargestellt werden.  
 
Tabelle 1: Ausschnitt des Beispieldatensatzes 

Planned 

treatment 

Age 

class Week 

No. of 

AEs 

Verum 20 0 0 
Verum 30 0 7 
Verum 40 0 26 
Verum 50 0 98 
Verum 60 0 205 
Verum 70 0 168 
Verum 80 0 62 
Verum 90 0 0 
Placebo 20 0 0 
Placebo 30 0 11 

Placebo 40 0 30 
Placebo 50 0 119 
Placebo 60 0 187 
Placebo 70 0 151 
Placebo 80 0 76 
Placebo 90 0 2 
Verum 20 1 0 
Verum 30 1 7 

 
Tabelle 2: Beschreibung des Beispieldatensatzes 

Variable Label Werte 

TRT01P Planned treatment Verum, Placebo 
AGENEW Age class 20, 30, 40, …, 90 
AEWK Week 0, 1, 2, 3, …, 26 
AVALS Number of AEs Number 

3.3 SAS-Programm 

Im Folgenden wird nun ein SAS-Programm erstellt, das die animierte Graphik erstellt. 
Zunächst wird in PROC TEMPLATE das Aussehen der einzelnen Bilder, die schließ-
lich animiert dargestellt werden sollen, definiert.  
 
PROC TEMPLATE; 
  DEFINE STATGRAPH ae_anim; 
    DYNAMIC _byval_; 
    BEGINGRAPH / designwidth=1200px designheight=960px; 
      ENTRYTITLE halign=center 'Week = ' _byval_; 
 
      LAYOUT DATAPANEL CLASSVARS=(trt01p) / headerlabeldisplay=value 
             columns=2 rows=1 rowaxisopts=(label='TEAE counts' 
             labelattrs=(size=18pt) tickvalueattrs=(size=16pt)  
             linearopts=(viewmin=0 viewmax=2500))  
             columnaxisopts=(label='Age at baseline [years]‚ 
             labelattrs=(size=18pt) tickvalueattrs=(size=16pt)  
             linearopts=(viewmin=20 viewmax=90)); 
 
        LAYOUT PROTOTYPE; 
          SCATTERPLOT x=agenew y=avals / group=trt01p  
               markerattrs=(size=15px symbol=circlefilled); 
        ENDLAYOUT; 
      ENDLAYOUT; 
    ENDGRAPH;    
  END;    
RUN; 

Zunächst kann die BY-Variable, welche die verschiedenen Graphiken klassifiziert, als 
dynamische Variable _BYVAL_ angegeben werden. Somit kann die entsprechende Va-
riable dann später im SGRENDER-Aufruf spezifiziert werden.  

Nach eventuellen globalen Einstellungen (hier die Breite und Höhe jeder Graphik) in 
BEGINGRAPH und Angabe einer Überschrift mit ENTRYTITLE wird das Aussehen 
jeder einzelnen zu erstellenden Graphik in einem LAYOUT DATAPANEL festgelegt. 
Hierbei wird die Behandlung als Klassenvariable bestimmt. Die einzelnen Optionen 
sind annähernd selbsterklärend, können aber natürlich in der SAS-Hilfe nachgeschlagen 
werden. 
 
Schließlich muss noch angegeben werden, von welchem Typ die einzelne Graphik sein 
soll. Die geschieht mit LAYOUT PROTOTYPE. In unserem Beispiel  wählen wir einen 
Scatterplot mit der Altersklasse auf der x-Achse und der Anzahl an AEs auf der y-
Achse. Mit Optionen kann auch hier das Aussehen des Plots verändert werden. 
Dieses Beispielprogramm kann nur einen ersten Einblick in die Möglichkeiten der Er-
stellung animierter Graphiken mit GTL geben, hilft aber hoffentlich, wenn notwendig, 
sich in die Materie einzuarbeiten. 
 
Nach der Definition des Layouts unserer Graphiken werden nun die notwendigen Opti-
onen angegeben, bevor PROC SGRENDER unser Werk vollendet.  
 
ODS LISTING GPATH='c:\documents\graphics'; 
OPTIONS NOBYLINE PAPERSIZE=('8 in', '6.4 in') PRINTERPATH=gif 
        ANIMATION=start ANIMDURATION=0.13 ANIMLOOP=yes NOANIMOVERLAY 
        NONUMBER; 
ODS PRINTER file='c:\documents\graphics\ae_anim1.gif'; 
ODS GRAPHICS / RESET=all WIDTH=8in HEIGHT=6.4in IMAGEFMT=png; 
 
PROC SGRENDER DATA=in.final_ae TEMPLATE=ae_anim; 
  BY aewk; 
RUN; 
 
ODS GRAPHICS / RESET=all; 
OPTIONS PRINTERPATH=gif ANIMATION=stop; 
ODS PRINTER close; 

 

Mit ODS LISTING GPATH wird der Pfad angegeben, in dem die Graphiken gespei-
chert werden sollen. Danach folgen die notwendigen Optionen, um eine animierte Gra-
phik erstellen zu können. Schließlich werden im PROC SGRENDER die einzelnen 
Graphiken sowie die Animation dieser erstellt.  

Wenn die entsprechenden Optionen bekannt sind, ist somit die Erstellung von animier-
ten Graphiken auch in GTL problemlos möglich. 

In unserem Beispiel werden 27 einzelne png-Dateien (für die Wochen 0 bis 26) erstellt 
sowie eine gif-Datei, die diese Dateien als Animation enthält. 

Abbildung 1: Finale Dateien im Beispiel 
 

In dieser Ausarbeitung kann die Animation leider nicht gezeigt werden, nur eine einzel-
ne der 27 animierten Graphiken soll hier noch gezeigt werden: 
 

 
Abbildung 2: Beispielgraphik für Woche 0 

4 Zusammenfassung 

Animierte Graphiken können auch mit GTL erstellt werden und das mit relativ wenig 
Aufwand. In einem PROC TEMPLATE kann durch Verwendung der Befehle LAY-
OUT DATAPANEL und LAYOUT PROTOTYPE das Aussehen der einzelnen Graphi-
ken in der Animation definiert werden. Es gibt verschiedene Optionen, mit denen die 
Animation den eigenen Wünschen angepasst werden kann. 
 
 
 


